```
OTA_STATE_VARIABLE
* dsauersanjose@aol.com  10/21/08
* www.idea2ic.com
*
*
*        R0   10K
*  ____/\  /\  /\_
* |      \/  \/   |          IABC= 1mA->0
* |  R1  1K       |
* |   _/\  /\  /\_|
* | _|_  \/  \/   |
* | \\\       ____| INP0        BP     R5 20K
* |          |                   _____/\  /\  /\_
* |____      |__|\     OUT0 |\  |       \/  \/   |INP1
*    _|_        | \/\/\_____| \_|                |___|\      OUT1|\   LP
*   /_  \    ___| /\/\/ _|_ | / |    _/\  /\  /\_|   | \/\/\_____| \__
*  // \  \  |   |/      ___ |/  |  _|_  \/  \/    ___| /\/\/ _|_ | /  |
*  \   \//  |  B_OTA0   _|_     |  \\\           _|_ |/      ___ |/   |
*   \___/   |           \\\ C0  |         R6     \\\ B_OTA1  _|_      |
*    _|_    |                   |         1k                 \\\ C1   |
*    ///    |  R2   20K  800PF  |                                     |
*           |_/\  /\  /\________|                           800PF     |
*           |   \/  \/                                                |
*       INN0|       20K  R3                                           |
*           |_/\  /\  /\______________________________________________|
*           |   \/  \/
*           |   1K      R4
*           |_/\  /\  /\_
*               \/  \/  _|_
*                       \\\
*

VIN       VIN        0     SIN(   0    10m   10000    )  AC 10m
R0        VIN        INP0  10K
R1        INP0       0     1k
B_OTA0    OUT0       0     I =   -1*v(VIABC)*tanh(( v(INP0) -v(INN0) )/.052)
C0        OUT0       0     800p
B_BUF0    BP         0     V =    v(OUT0)
R2        INN0       BP    20k
R3        INN0       LP    20k
R4        INN0       0     1k
R5        BP         INP1  20k
R6        INP1       0     1k
B_OTA1    OUT1       0     I =   -1*v(VIABC)*tanh(( v(INP1)   )/.052)
C1        OUT1       0     800p
B_BUF1    LP         0     V =    v(OUT1)
V_Iabc    VIABC      0     PWL    ( 0 1m 1m 0 )

.control
tran      1u         1m   0      1u
set       pensize =  2
plot      v(vin)     v(bp)       v(lp)
op
ac        dec        10    1      1000K
plot      db(bp)     db(lp)     title Gain2OUT@IABC1000uA
alter     V_Iabc     dc =         100u
ac        dec        10    1      1000K
plot      db(bp)     db(lp)     title Gain2OUT@IABC100uA
alter     V_Iabc     dc =         1u
ac        dec        10    1      1000K
plot      db(bp)     db(lp)     title Gain2OUT@IABC1uA
.endc

.end
```

=====================END_OF_SPICE============================

To Covert PDF to plain text click below
http://www.fileformat.info/convert/doc/pdf2txt.htm
This code works with winspice.



FIGURE 14. Voltage Controlled State Variable Filter

The state variable filters have the beauty to them that the feedback independently defines the shape of the filter. The following two sites show the feedback level differences between Bessel and Butterworth filters.

http://www.idea2ic.com/PlayWithSpice/pdf/Bessel_6P_State_Variable_txt.pdf
http://www.idea2ic.com/PlayWithSpice/pdf/Butterworth_6P_State_Variable_txt.pdf

Independently, the gm and C stages define the frequency response. This is seen when the value of IABC is adjusted. At 1mA the response is..



At a thousand times less current, the frequency response is a thousand times lower.



Assuming the OTAs and capacitors and buffers form perfect integrators, there are two different important feedbacks. The feedback from R2 has 90 degrees of phase shift while R3's feedback has 180 degrees of phase shift.

```
*
*        R0   10K
*  ____/\  /\  /\_
* |      \/  \/    |          IABC= 1mA->0
* |  R1  1K       |
* |   _/\  /\  /\_|
* | _|_  \/  \/   |
* | \\\       ____| INP0        BP     R5 20K
* |          |                   _____/\  /\  /\_
* |____      |__|\     OUT0 |\  |       \/  \/   |INP1
*    _|_        | \/\/\_____| \_|                |___|\      OUT1|\   LP
*   /_  \    ___| /\/\/ _|_ | / |    _/\  /\  /\_|   | \/\/\_____| \__
*  // \  \   |   |/      ___ |/  |  _|_  \/  \/    ___| /\/\/ _|_ | /  |
*  \   \//  |  B_OTA0   _|_     |  \\\           _|_ |/      ___ |/   |
*   \___/   |           \\\ C0  |         R6     \\\ B_OTA1  _|_      |
*    _|_    |                   |         1k                 \\\ C1   |
*    ///    |  R2   20K  800PF  |                                     |
*           |_/\  /\  /\________|                           800PF     |
*           |   \/  \/                                                |
*       INN0|       20K  R3                                           |
*           |_/\  /\  /\______________________________________________|
*           |   \/  \/
*           |   1K       R4
*           |_/\  /\  /\_
*               \/  \/  _|_
*                       \\\
*
```

The ratio of 90 degree to 180 degree phase shift feed back obviously sets the stability of the filter. Think of the value of R2 as setting the Q of the filter. The following web site shows that in more detail.

http://www.idea2ic.com/PlayWithSpice/pdf/State_Variable_s_eq_1_txt.pdf

Whenever the integrators of the filter change, the frequency response scales by the same amount.

http://www.idea2ic.com/PlayWithSpice/pdf/State_Variable_f_1KHz_txt.pdf



For filters using OTAs, this corresponds to gm an C terms. The above curves show IABC going from 1mA to zero. More information is given below.

http://www.idea2ic.com/PlayWithSpice/pdf/State_Variable_OTA_1KHz_txt.pdf

All state variable filters three outputs. The high pass not shown is just the input signal minus the lowpass.



This filter is using 800pF capacitors. Running at 1uA, the filter is center at a few hundred hertz. A BiCMOS process can easily run down at the 1nA level. And the buffers can use CMOS inputs that have an input impedance of glass. That means the same filter could be built using capacitors 1000 times smaller. That means audio frequency filters can be put completely on chip.